
0.8 ジュニア版     データのグラフ化

　今までは関数を使う、または、自ら作成してデータを分析することのさわりに取り組み

ましたが、ここではデータをグラフ化することをやってみましょう。まずは、時系列デー

タをプロットすることから始めて、平均の応用である移動平均というものをとってそれら

をプロットさせましょう。また、時系列ではない度数を見るグラフとして、ヒストグラム

利用して度数分布や累積度数分布を見てみましょう。最後に、ヒストグラムをならしてス

ムースにしたようなカーネル密度関数について少しふれましょう。

時系列データ

　原始的に Copy&Paste だけで株価データをGAUSSのプログラム上に取り入れるには次の

ようになります。

【手順】

１） Yahoo Japan等で公開されている株価の時系列をカーソルで黒くして Copyする

２） Excelなどのワークシートソフト上に Pasteする

３） いくつかある系列のうち、終値の列だけを選択して Copyする

４） もう 1枚ワークシートを開いてそこに Paste する

５） 終値１列のデータをテキスト形式で保存する

６） 保存されたデータを開いて Copy&Pasteして GAUSSのエディター上で扱う

以下では、まずデータをこれまでと同じようにプログラム上で読み込んでから、この場合

一番上の行が最近で、一番下が一番遠い過去になっていますから、順序を逆にします。そ

うしてから、ｘ軸を任意の続き番号（例えば１から50まで）、ｙ軸をこれら株価の終値とし

て、ｘｙグラフで描いてみます。

プログラム

new;

cls;

let data[50,1]=

      60

      61

      60

      62

      60

      63

      64

      65

      69



      70

      69

      66

      64

      64

      62

      64

      65

      55

      53

      57

      60

      58

      63

      63

      63

      68

      69

      69

      68

      69

      70

      67

      66

      67

      68

      69

      69

      77

      80

      80

      78

      76

      76

      72

      71



      70

      73

      71

      78

      80

;

data=rev(data);

library pgraph;

graphset;

xy(seqa(1,1,50),data);

グラフ表示

プログラム各行の説明

１行目　メモリを初期化する

２行目　画面をクリアする

３～５４行目　変数dataに50×１のディメンションでセミコロンまでのスペースに書かれ

　　　　ているデータを取り込んで入れる

５５行目　組込み関数 revを用いて、変数 dataを上と下を逆にする



５７行目　グラフを描く機能の入った pgraphというライブラリを呼び出す

５８行目　ライブラリ pgraphの各種設定のグローバル変数を初期化する

５９行目　ｘ軸に１から１ステップの 50 個のシークエンスを割り当て、ｙ軸には変数 data

　　　　　の内容を割り当てて、ｘｙグラフで描く

前半部分のプログラム上でのデータの取り込みはこれまでと同じ形式です。株式データな

どは直近のものが上にきていることが多いので、その場合には、上（列の１行目）が古く

て、下（列の最終行）が新しくするように上下を反転させるのには、組込み関数 revを使い

その中に data を入れて変換されたものをあらためて data と置き直します（置き直す変数

名は違うものでもかまいません）。各種グラフを描くには、GAUSS では通常の組込み関数

では対応していなくて、外部のライブラリという形の組込み関数群を呼び出して利用可能

な状態にしておいてから、例えばｘｙグラフなどの描画を行ないます。少し面倒ですが、

     library pgraph;

     graphset;

の２つの命令が必要なだけです。pgraphという名前のグラフ描画関係の組込み関数がたく

さんはいったライブラリを呼び出して利用可能にして、次にその各種設定に使われるグロ

ーバル変数と呼ばれるものを初期化します。これだけです。この２つは覚えておいてくだ

さい。これらがなければｘｙグラフ他各種のグラフを書こうとしても動かないか、または

画面が意図しないものになります。最後に、ｘｙグラフで、その第１要素にｘ座標となる

べき列ベクトルを入れます。ここでは仮に１から 50 までのシークエンスとしましょう。そ

の第２要素にはｙ座標となる変数 dataの内容を入れます。GAUSSは自動描画させるパッ

ケージソフトウエアではなくて、設定もすべて自分でしなければなりません。ｘ軸とｙ軸

の長さ（それぞれの列の行数）は必ず一致する必要があります。この場合 50で共通です。

以上がグラフの中心となるｘｙグラフの描き方の基礎となる部分です。

次に、上の時系列グラフに基準線を入れてやりましょう。例えば、70という上限線と 53

という下限線をグラフに書き入れたいときには、

     xy(seqa(1,1,50),data~70*ones(50,1)~53*ones(50,1));

50×１のすべて１からなる列ベクトルを利用して、それに任意の数をかけてやることで、

同じ数からなる列ベクトルを作成します。そうして作成した列ベクトルをｙ軸側に２列目

の変数、３列目の変数…という具合に水平方向に～の記号でマージしてやれば、もとのデ

ータも含めて３列からなるｙ軸側のデータ行列ができます。これによって、ｘ軸側は共通

として、３つの系列のグラフが一度に同じ画面に描かれます。基準線は１本であっても、

はたまた３本以上であっても同じ要領でできます。



グラフ表示

　株価グラフでよく使う５日間および２５日間移動平均を計算して、基準線と同じように

元データに水平方向にマージすることでグラフ表示してやりましょう。

プログラム

（　同上の部分　）

data=rev(data);

library pgraph;

graphset;

xy(seqa(1,1,50),data~ma5(data)~ma25(data));

proc ma5(x);

   local m,i,j;

   m=zeros(rows(x),cols(x));

   i=5;

   do while i<=rows(x);

      j=1;

      do while j<=5;

         m[i,.]=m[i,.]+x[i-5+j,.];



         j=j+1;

      endo;

      i=i+1;

   endo;

   m=m/5;

   m[1:4,.]=miss(m[1:4,.],0);

   retp(m);

endp;

proc ma25(x);

   local m,i,j;

   m=zeros(rows(x),cols(x));

   i=25;

   do while i<=rows(x);

      j=1;

      do while j<=25;

         m[i,.]=m[i,.]+x[i-25+j,.];

         j=j+1;

      endo;

      i=i+1;

   endo;

   m=m/25;

   m[1:24,.]=miss(m[1:24,.],0);

   retp(m);

endp;



グラフ表示

ボリンジャーバンド

移動平均の応用として、単純移動平均のまわりにσを求めて、例えば 20期の移動平均に

対して逐次的に

　　　Upper Band = MA + ２σ

　　　Lower Band = MA – ２σ

なる上限と下限を求めるものです。ここで、σは GAUSS の標準機能のｎ－１で割るもの

ではなくて、ｎで割るバージョンを使うことにします。

プログラム

（　同上の部分　）

data=rev(data);

library pgraph;

graphset;

xy(seqa(1,1,50),data~Bollinger(data));

proc Bollinger(x);

   local m,i,j,sig;

/* MA20 calculation */



   m=zeros(rows(x),cols(x));

   i=20;

   do while i<=rows(x);

      j=1;

      do while j<=20;

         m[i,.]=m[i,.]+x[i-20+j,.];

         j=j+1;

      endo;

      i=i+1;

   endo;

   m=m/20;

   m[1:19,.]=miss(m[1:19,.],0);

/* sigma calculation */

   sig=zeros(rows(x),cols(x));

   i=20;

   do while i<=rows(x);

      sig[i]=sd(x[i-19:i,.]);

      i=i+1;

   endo;

   retp(m~(m+2*sig)~(m-2*sig));

endp;

proc sd(x);

   local n,sum,i,xbar,s2,s;

   n=rows(x);

   sum=0;

   i=1;

   do while i<=n;

      sum=sum+x[i,.];

      i=i+1;

   endo;

   xbar=sum/n;

   s2=sumc((x-xbar)^2)/n;   @ Divided by n. @

   s=sqrt(s2);

   retp(s);

endp;



グラフ表示

上のプログラムの procedure の前半はこれまでの５期と２５期の移動平均のプログラムを

20に修正しただけです。後半は、その下に置いている sdというｎで割るバージョンのσを

求める procを用いて、ｉ－19からｉまでの 20データのσを逐次的に計算したものです。

最後に、移動平均の系列ｍに２倍したσを足し引きしたものをマージして、移動平均と上

限と下限の３列の変数行列をリターンとして外部に返しています。冒頭で、もとの系列デ

ータとマージして、合計４つの系列をｘｙグラフで表示させています。

度数分布

プログラム

new;

cls;

data={90,62,38,81,55,40,78,100,79,65,

      60,97,62,83,70,48,90,58,74,68,

      75,60,48,70,95,65,77,55,40,73};



library pgraph;

graphset;

v={30,40,50,60,70,80,90,100};

call hist(data,v);

グラフ表示

上のヒストグラムは縦軸が度数ですが、パーセントにするには hist のかわりに histp を使

い全く同じようにグラフ化します。ｖのところを各階級の区切り値からなるベクトルでは

なくて、ただ単に７とか８とか数値を１つ入れると、その数の階級からなるヒストグラム

が自動的にグラフ化されます。

プログラム

new;

cls;

data={90,62,38,81,55,40,78,100,79,65,

      60,97,62,83,70,48,90,58,74,68,

      75,60,48,70,95,65,77,55,40,73};

library pgraph;



graphset;

pqgwin many;

v={30,40,50,60,70,80,90,100};

{b,m,freq}=hist(data,v);

histf(cumsumc(freq)./sumc(freq),v);

グラフ表示

カーネル密度

プログラム

new;

cls;

data={90,62,38,81,55,40,78,100,79,65,

      60,97,62,83,70,48,90,58,74,68,

      75,60,48,70,95,65,77,55,40,73};



call plotdens(data);

proc plotdens(x);

   local n,h,points,kern,i;

   n=rows(x);

   x=sortc(x,1);

   h=1.059*stdc(x)*n^(-0.2)*1.740;　　　@ One of the ways to get a right band h @

   points=seqa(x[1],(x[n]-x[1])/100,101);

   kern=zeros(101,1);

   i=1;

   do while i<=101;

      kern[i]=ukernel(points[i],x,h);

      i=i+1;

   endo;

   library pgraph;

   graphset;

   xy(points,kern);

   retp(kern);

endp;

proc ukernel(x0,x,h);

   local u,kv;

   u=(x0-x)/h;

   kv=0.5*(abs(u).<=1);

   retp( sumc(kv/(rows(x)*h)) );

endp;

グラフ表示



練習問題

【問１】

Yahoo Japan等で百円前後の株価の銘柄について、最初に示した取り込み方で GAUSS上

に置いた上で、そのグラフを描こう。

【問２】

都道府県別になっている政府統計を何か使って、相対度数分布と累積度数分布の２つのグ

ラフを描いてみよう。

【問３】

カーネル密度の計算では簡単化のためフラットな一様カーネルが用いられた。これをガウ

スカーネルで描くとどうなるだろうか（ヒント：一番下に置く ukernel は下記の gkernel

に変更する。また、1.740の定数がついているところは、1.740を１に変更する）。

proc gkernel(x0,x,h);

   local u,kv;

   u=(x0-x)/h;



   kv=pdfn(u);

   retp( sumc(kv/(rows(x)*h)) );

endp;

発展

